
;-----------------------------------------------------------

; cav.dat

; Undrained cylindrical cavity expansion in Cam-Clay medium

;-----------------------------------------------------------

title

   Undrained cylindrical cavity expansion in Cam-Clay medium

config fluid

; --- model geometry ---

gen zone brick p0 1.0     0.0       1.0   p1 200.0    0.0       1.0   &

               p2 1.0     0.0       0.0   p4 200.0    0.0       0.0   &

               p3 0.9877  0.1564    1.0   p6 197.5377 31.2869   1.0   &

               p5 0.9877  0.1564    0.0   p7 197.5377 31.2869   0.0   &

               size 31 1 1 ratio 1.1 1 1

def find_id

   pnt1 = gp_near(1.,0.,1.)

   pnt2 = gp_near(1.,0.,0.)

   pnt3 = gp_near(0.9877,0.1564,1.)

   pnt4 = gp_near(0.9877,0.1564,0.)

   id1  = gp_id(pnt1)

   id2  = gp_id(pnt2)

   id3  = gp_id(pnt3)

   id4  = gp_id(pnt4)

   pntz = z_near(1.,0.07,0.5)

cc   = cos(4.5*degrad)

   ss   = sin(4.5*degrad)

sc   = 2.*cc*ss

   cc   = cc*cc

ss   = ss*ss

   b_vel= 1.e-5

b_vx = 0.9877*b_vel

   b_vy = 0.1564*b_vel

end

find_id

range name out cyl end1 0 0 -10 end2 0 0 10 rad 197.5377 not

range name in  cyl end1 0 0 -10 end2 0 0 10 rad 1.001

; --- model properties ---

mo cam-clay

prop shear 74.

prop mm 1.2 lambda 0.15 kappa 0.03 mp1 1.0 mv_l 2.3

mo fl_iso

; --- boundary conditions ---

fix y z range y -.001 .001

apply nvel 0.0 plane dip 90.0 dd -9.0 origin 0.0 0.0 0.0 &

  range in not y 0.07 200.0



apply dvel 0.0 plane dip 90.0 dd -9.0 origin 0.0 0.0 0.0 &

  range in not y 0.07 200.0

apply nstress -1.65 range out

fix x y z range in

ini xvel b_vel range id id1 id1

ini xvel b_vel range id id2 id2

ini xvel b_vx yvel b_vy range id id3 id3

ini xvel b_vx yvel b_vy range id id4 id4

ini sxx -1.65 syy -1.65 szz -3.

; model settings ---

set fluid biot on

set fluid off

set large

; --- fish functions ---

; ... initial specific volume, tangent bulk modulus, porosity ...

def set_prop

   pnt = zone_head

   s1 = -z_sxx(pnt)

   s2 = -z_syy(pnt)

   s3 = -z_szz(pnt)

   p0 = (s1 + s2 + s3) / 3.

   q0 = sqrt(((s1-s2)*(s1-s2)+(s2-s3)*(s2-s3)+(s3-s1)*(s3-s1))*0.5)

e0 = q0 / (z_prop(pnt,'mm') * p0)

   pc = p0 * (1. + e0 * e0)

p1 = z_prop(pnt,'mp1')

   vc = z_prop(pnt,'mv_l') - z_prop(pnt,'lambda') * ln(pc/p1)

   v0 = vc - z_prop(pnt,'kappa') * ln(p0/pc)

   b0 = v0 * p0 / z_prop(pnt,'kappa')

   n0 = (v0 - 1.) / v0

   bb = 10. * b0

   bi = 100. * b0

   command

      prop mpc pc bulk_bound bb

      ini biot_mod bi

   end_command

end

; ... numerical variables ...

def path

   srad = z_sxx(pntz)*cc+z_syy(pntz)*ss+z_sxy(pntz)*sc

   p_fl = z_pp(pntz)

   c_tp = -(z_sxx(pntz)+z_syy(pntz)+z_szz(pntz)) / 3.

   c_ep = z_prop(pntz,'cp')

   c_q  = z_prop(pntz,'cq')

   c_cq = c_ep * z_prop(pntz,'mm')



   c_v  = z_prop(pntz,'cv')

   c_a  = 1. + gp_xdisp(pnt2)

   c_b  = z_prop(pntz,'bulk')

   c_pc = z_prop(pntz,'mpc')

end

; --- histories ---

hist nstep 500

hist path

hist srad

hist p_fl

hist c_a

hist c_ep

hist c_cq

hist c_tp

hist c_v

hist c_q

hist c_b

hist c_pc

hist unbal

; --- test ---

set_prop

step 100000

save cav.sav


